Beginning Scripting

For 

TWX Proxy

Part 1

by Promethius

As a beginner to programming and to TWX proxy, the concept of writing complicated scripts can be overwhelming.  Without programming experience, one would think that it would be extremely hard; however, that is not the case.  With some basic information, you will be able to write TWX scripts that are at your level of game play.

To write more complicated scripts, requires a very good understanding of the game since you are essentially automating the keystrokes that you would normally enter.  What you know, is what your script has to know in order to carry out the desired function.

For this document I am using // to denote comments at the end of a line.  The text following the // is not part of the code.  Above line comments are denoted by the use of #.

A quick disclaimer regarding the information I am providing.  I am not a programmer, and there may be better more efficient methods of doing what I am showing.  I am attempting to write for example, and not necessarily for efficiency.  I will try and do both, but there will be times when I am just showing a way of using a command.

Before getting into the commands used in TWX scripting we need to figure out how we can design a simple script.  The design of a script will depend heavily on your knowledge of Trade Wars.  The commands are straight forward, but it is the "what ifs" that have to be taken into account.  The script will only do  what we tell it to do, and it will only know as much as we tell it before it starts running.

Goal oriented programming can be viewed as follows:

Goal:  I want to go from my home to the store.

1. I need my car keys.

2. I need money

3. Car starts

4. Car has gas

5. Back out of drive

6. Head for store

7. Get to store

Fairly straight forward, but we made assumptions to get us where we wanted to go.  Scripts cannot make assumptions, only what we or the game tells it and that we have scripted for it to accept.

To be a pain in the goal above:  We assumed we were dressed, we opened the door, we walked to the car, we opened the car door, plus several others.

Similarly, in TWX you can write (using paper or your favorite editor) out what it is you want to do and then the things that will be required to get it done, and you can do it without writing script.

Goal:  I want to t-warp to sector 1

Do I have enough turns
if not then halt


Make sure I have t-warp capability
if not then halt


Do I have enough ore?

               if not then is there a port selling it or a planet I can get ore from?
           
            Can I t-warp direct to 1 with my alignment

   have to check to see if alignment allows direct t-warp

  If I can warp direct then go

                 if not, can I warp to another sector adjacent to fed?


    do I have a fighter down in the adjacent sector?

if I can warp adjacent, then check for t-warp lock, 

if lock is made go and then move to 1

Yes, that is kind of crude, but there are factors you have to consider so you don't end up in the daily logs showing that you fused the ship that took you 3 hours of PPT to get.

Editors for Scripting

TWX scripts are created using any ASCII text editor including Notepad, and ConText.  Notepad can be used very easily for quick uncomplicated scripting, but you will want to find an editor with more features.  ConText and EditPad are two that I have used a lot.  

ConText has a TWXproxy file that allows key words to be highlighted in color and makes debugging somewhat easier.

ElderProphet has an editor designed specifically for TWX scripting that would be something to look into.

TWX Scripts .TS vs .CTS

By now you have probably seen scripts with the extension of .TS and .CTS, and since you are reading this you probably already know the .TS is the non compiled source version of a script, and that .CTS is the compiled version.  There seems to be no difference in the performance of a script that is compiled verses uncompiled source version.

TWX Database

When Xide created TWX Proxy, he had the kindness to create a specialized database that scripters could easily pull information from.  There are helpers such as SWATH that can be used in a similar fashion.

Information that can be retrieved is shown in the help file that comes with TWX.  To see this information, right click on the TWX icon and select Help → Scripting Reference.  The reference also lists all commands and the syntax that is used.  I will cover some of those items that I commonly use.

Variables

Variables are containers for numbers and text information.  Typical uses include the menu choices you make to set up options the script will use, and for containing any information that you may need to pass on to other parts of the script.

Variables are only local to the script they are running in and cannot be passed directly to other scripts.

To set a variable we use the setVar command:


setVar $i 10

//  Number


setVar $i ″attack"
//  Text


setVar $i TRUE 
//  Boolean (True/False)

As stated above, you cannot pass a variable directly to another script, but you can save the variable so that another script can read it.


setVar $i 10
// we've set our variable's value


saveVar $i
// now we've saved it to the database .cfg file (located under \twx\data)

Now that we've saved the variable, we need our second script to read it:


loadVar $i
// typically we want to have descriptive variable names rather than $i

You can view active variables in a script with $sv and this will help in debugging.  If you have an array, it will list the array values – maybe all 20,000 if you have an array of sectors setup in a 20k sector game.

Arrays

Arrays can be viewed as variables with a very distinct advantage, multiple setting of the variable (array) by using a looping structure such as while/end.

Arrays are of two types:  static, where the upper limit is set, and dynamic where the array grows as needed during the script run.  Static arrays are considered to be faster than dynamic arrays.


setArray $myArray 100      // setArray is use to create a static array of 100 elements





        //  the elements of the array are initialized to zero.


Static Array example


setArray $myArray 100
// Initialize our array to 100 element equal to zero.


setVar  $myCounter 1

// Setup a normal variable to be used as a counter.


while ($myCounter <= 100)
// A while/end is a looping structure that will be covered later

             # The line below sets the array value in a similar way to a normal variable.  The exception


 # is that the array has brackets around the counter.  The value of $myCounter will be store


 # into the array at the counter position.  To visualize this, if $myCounter = 5, then the 


 # line would evaluate to:  setVar $myArray[5] 5, where 5 is stored into the fifth element.


     setVar $myArray[$myCounter] $myCounter


     add $myCounter 1

            end

A dynamic array uses the same code, however you do not use the ″setArray $myArray 100″ that the static array example uses.

Cut & paste:


setArray $myArray 100


setVar $myCounter 1


while ($myCounter <= 100)


     setVar $myArray[$myCounter] $myCounter


     add $myCounter 1

            end

            # reset $myCounter and echo to client


setVar $myCounter 1


while ($myCounter <= 100)


     echo "*" ANSI_12 $myArray[$myCounter]

                 add $myCounter 1

             end

Decision Structures

  If/End        If/ElseIf/End

  The "IF" control structure is basically looking to see if something  is true or false.  Sometimes if something is false you want to do what comes next.

      # note:  // end of line comments are not currently supported by twx

     setVar $x 1        


// sets the variable $x to 1

     if ($x = 1)       


// we want to know if the value in the variable $x = 1

        echo "*" $x     


// echos to the client screen (* is the enter command)

     end                


// we are done with our check

     getRnd $x 1 10      

// get a random number in the range of 1 to 10

     if ($x = 1)

        echo $x

     elseif ($x = 10)      

 // see if it was 10

        echo "*Value of $x was 10"

     else               


 // value was neither of the above

        echo "*Value was neither 1 or 10"

        echo "*$x = " $x

     end

If decision/control structures will be found through all scripts to determine what actions are to be taken.

While/End

 Loops are a common method of doing something a certain number  of times or until something changes value in the loop.

Endless loops are a very bad thing that can easily max your cpu, and besides, you don't do anything except spin in a circle.  You will have to kill TWX to get out of it. (ctrl-alt-delete and then select TWX to kill).

     setVar $x 0       

// set the variable $x to 0

     while ($x < 10)   

// while $x is less than 10 loop

         add $x 1      

// add 1 to $x otherwise this is an endless loop

         echo "*" $x   

// echo the value of $x to the client

     end               

// end of while

     setVar $x 0

     while ($x < 10)

         add $x 1

         if ($x = 7)

            echo "*$x reached the value of 7"

         end

     end

The positioning of your add $x 1 can affect your output.  In the  first example of while ($x < 10), it  displayed values of 1 - 10.  The  next will show values of 0 - 9.

     setVar $x 0

     while ($x < 10)

       echo "*" $x

       add $x 1

     end

If you wanted values of 0 - 10 to display in the above example, you could change "while ($x < 10)" to "while ($x <= 10)".

     # putting a while loop to use

     # echo uses ANSI_12 to output the echo in red

     # CURRENTSECTOR gets our current sector from TWX

     # if $i is initialized to 1, because

     # sector.warps[CURRENTSECTOR][0] would not be valid.

     setVar $warpCnt sector.warpcount[CURRENTSECTOR]    
// get the warpcount

     setVar $i 1                                        



// init our var to 1

     while ($i <= $warpCnt)                             


// setup while loop value

       echo ANSI_12 "*" sector.warps[CURRENTSECTOR][$i] 
// display adjacent warps

       add $i 1                                         



// increment counter

     end                                                




// end of while loop

     halt                                               




// stop script

Cut & paste:

     setVar $warpCnt sector.warpcount[CURRENTSECTOR]    

     setVar $i 1                                        

     while ($i <= $warpCnt)                   

       echo ANSI_12 "*" sector.warps[CURRENTSECTOR][$i] 

       add $i 1                                         

     end                                               

     halt                                               

Outputting to the Client

Outputting to the client, and not the game can provide information that you want, and not put it out in the game.  There are two basic methods for doing this, echo and clientmessage, and one a bit more useful in TWX windows.

Echo

The echo command is the simplest method of outputting to the client.

echo "*hello world"    



// sends an enter command and hello world

The echo display can contain ANSI escape sequences and also the ANSI_xx constants to add color to the echo.

echo ANSI_12 "*hello world" 


// displays text in red

Variables can also be sent to the client with the echo command:

echo ANSI_12 "*hello world " $someVariable ANSI_11 " changed color"

Clientmessage

The client message command display a message similar to the echo command, however; its use is also more complicated and cannot use ANSI colors or the ANSI constants.  To display a message containing variables, you must concatenate the output with the ampersand (&).

clientmessage "hello world"

clientmessage "hello world " & $someVariable & " we can't change color"

TWX Windows

Many times you cannot read the text on the screen fast enough to pick out needed information due to the speed it comes in and scrolls.  A very good way to overcome this is to have your script output key information to a TWX window.  When gridding, it is good to know if your script detected an L5+ planet adjacent to your sector or alien sectors.  Information in the windows is limited only to what you think you may need.  The progress of a script can be monitored in a window to show remaining ore on a planet when colonizing as one example.

Creating TWX windows is a fairly simple thing to do so:

  setVar $hitSector "1432"                          // setup bogus hit

  setVar $nearFig "1232"                            // setup bogus nearest fig

  Window myWindow 300 275 "TWX Window Example" ONTOP

  #                      |             |      |                     |                              |- set the window on top

  #                      |             |      |                    Window Title

  #                      |             |     Vertical size of the window

  #                      |             Horizontal size of the window

  #                     Name to use to reference our window to our information in

  setVar $windowContent "  Fighter hit:  " & $hitSector & "*"  
// setup var for the window

                                                             



// & used to join the information

  # adding another line to our windowContent var - you have to end it with a "*"

  setVar $windowContent $windowContent & "  Nearest fig: " & $hitSector & "*"

  setWindowContents myWindow $windowContent    /

/ set the window contents to our var

#                                          |                    |- Variable

#                                         Name we set for the window above

  waitfor "³Turns"    



// waiting on you to hit / (thanks to LoneStar for that tip)

Yes, you can scroll your data as it comes in, but it requires a bit more work.  To scroll it back, requires a lot more work.  You will have to setup arrays to have it scroll and then manipulate the array position.  Since I am keeping this as basic as possible for people new to scripting/programming it is a bit beyond that I think.  .

Cut & Paste

  setVar $hitSector "1432"

  setVar $nearFig "1232"

  Window myWindow 500 275 "TWX Window Example" ONTOP

  setVar $windowContent "  Fighter hit:  " & $hitSector & "*"

  setVar $windowContent $windowContent & "  Nearest fig: " & $hitSector & "*"

  setWindowContents myWindow $windowContent

  waitfor "³Turns"

Let's Create Something

I will show an example of a photon adjacent sector that uses triggers and the TWX database.

Pseudo Code 

Goal:  Launch a photon into an adjacent sector on a fighter or limpet trip.


If I have photons continue else give a message and halt.

            get my current sector

            get the adjacent sectors ( an assumption is that they are figged and/or limpeted.)

            setup activation triggers

            wait for limpet or fighter trip


send photon into hit sector

# script name:  photonAdj.ts

# written by:
Promethius

# Purpose:
Photon adjacent sector on fighter/limpet trip

# Written:        Nov. 2008

# Need to check to see if we have photons.  This could come from the "i" command for trader
#  information or from the "/" command for quick stats.  A good quick stats routine is used in a lot of
# scripts and should be created and saved so it can be included in your scripts.   A problem with the "i"
# command is that if you do not have photons, the text will not appear.  You would have to do a work
# around for that but since this is for learning, we will do one possible work around as it may be useful
# in other scripts.

send "i"

# ok, using setTextLineTrigger so it reads the entire line and return

# setTextTrigger might only read part of the line and miss the text we are looking for

setTextLineTrigger  numberPhotons :numberPhotons "Photon Missiles: "

# now for a bit of a kludge work around – there is another way to do this w/o a delay trigger:

setDelayTrigger noTextFound :noTextFound 2000

# The delay trigger will activate in 2,000 ms (1,000 ms = 1 second) so we will know if we do not have 

# photons

# now to pause and wait on our text trigger or delay trigger to activate.

pause

# our "goto" for the delay trigger 

:noTextFound

  echo ANSI_12 "!!! Might be nice to have photons for this – we DON'T. !!!"

  halt

# our text trigger fired so we have photons on board our ship.

:numberPhotons

# when using triggers it is good to kill them when they are no longer needed.  With a delay trigger, you have to kill the trigger so it does not activate  on you and cause bad things to happen.

killTrigger noTextFound

# you may or may not want the number of photons, if you don't you could leave out the following 
# line and just let the script continue.  We are telling the script we want the 3rd word stored into a variable ($numPhotons) from the line that our text trigger activated on.

  getWord CURRENTLINE $numPhotons 3

# The pseudo code said we need to get our current sector and there are several ways to do that, but

# the easiest way starting with TWX 2.04 is to use CURRENTSECTOR so that is what we will do.

# Now we need our adjacent sectors and since those are stored in our database, that is where we

# will get them.

# First we need to get the number of adjacent sectors.

setVar $warpCnt SECTOR.WARPCOUNT[CURRENTSECTOR]

# Now we can setup our adjacent sector triggers, but we don't know them?  This is one place a while 

# loop can save you a bit of typing so bear with me.

setVar $i 1

while ($i <= $warpCnt)

   setTextLineTrigger limpetLbl & $i :limpHit "Limpet actived in sector " & sector.warps[currentsector][$i]

   setTextTrigger figLbl & $i :figHit "Deployed Fighters Report Sector " & sector.warps[currentsector][$i] & ":"

  add $i 1

end

# The above example sets triggers for specific sectors and can help prevent spoofing of a script.

# The labels for each trigger is set with a concatenation with the loop (limpetLbl2 as an example.)

# You should read "stupid ansi tricks" in the TW-Cabal link from NavHaz.com (http://tw-cabal.navhaz.com/strategy/twxansitricks.html)   for ways to prevent

# getting spoofed.  This script does not have any prevention added.

pause

:limpHit

  killalltriggers

  getword currentline $hit 4

  goto :sendHeat

:figHit

  killalltriggers

  getword currentline $hit 5

:sendHeat

# cpy (computer/photon/yes) / $hit, the sector they tripped / "*  y"  enter and "y" (yes I want to send it)

send "cpy" $hit "*y  q"

Cut & Paste

# script name:  photonAdj.ts

# written by:
Promethius

# purpose:
Photon adjacent sector on fighter/limpet trip

# written:        Nov. 2008

# to add:  Spoof checks and verify that it wasn't an alien that tripped the fighter

#              Actions to perform after the photon is fired.

send "i"

setTextLineTrigger  numberPhotons :numberPhotons "Photon Missiles: "

setDelayTrigger noTextFound :noTextFound 2000

pause

:noTextFound

  echo ANSI_12 "!!! Might be nice to have photons for this – we DON'T. !!!"

  halt

:numberPhotons

  killTrigger noTextFound

  getWord CURRENTLINE $numPhotons 3

  setVar $warpCnt SECTOR.WARPCOUNT[CURRENTSECTOR]

  setVar $i 1

  while ($i <= $warpCnt)

      setTextLineTrigger limpetLbl & $i :limpHit "Limpet actived in sector " & sector.warps[currentsector][$i]

    setTextTrigger figLbl & $i :figHit "Deployed Fighters Report Sector " & sector.warps[currentsector][$i] & ":"

    add $i 1

  end

   pause

:limpHit

  killalltriggers

  getword currentline $hit 4

  goto :sendHeat

:figHit

  killalltriggers

  getword currentline $hit 5

:sendHeat

send "cpy" $hit "*y  q"

# I would suggest viewing the trigger settings of the above script so you can see how they are being set # up.  Use the $st command to view your triggers.  The trigger view and the variables that can be

# viewed can help when trying to debug a script.

Using Script Triggers

Yes, we have already been using them in the examples and they may not have made much sense so we will remedy that now.

A trigger can be viewed as a mechanism that only runs when data appears that the trigger is looking for or, in the case of a delay trigger, a certain time setting has expired.  A delay trigger can be used to determine if something did not happen in a certain period of time and branch to another part of the script.

setTextTrigger  triggerLabel :triggerGoTo "Deployed fighters report"

                                    |                         |                               |- The text we are waiting on

                                    |                         The label we will go to when the trigger fires

                                 The label that names this trigger so we can kill it if we want

A limitation of the setTextTrigger is that it fires as soon as the text it is looking for comes in.  In the above example, you may or may not get additional information from the line that cause the trigger to activate. 

setTextLineTrigger  triggerLabel :triggerGoTo "Deployed fighters report"

                                        |                         |                               |- The text we are waiting on

                                        |                         The label we will go to when the trigger fires

                                    The label that names this trigger so we can kill it if we want

The setTextLineTrigger works like the setTextTrigger with the exception of it reads a full line that has to have a line return at the end.  It would not read the sector command line until you enter a command that causes a new line.  The setTextTrigger could be set to read the sector command line since it activates  as soon as the text it is looking for arrives.

setDelayTrigger triggerLabel :triggerGoTo  3000

                                        |                |                |- Time in milliseconds to wait before firing (1000 ms = 1 second)

                                        |                The label we will go to when the trigger fires

                                    The label that names this trigger so we can kill it if we want

The require a "pause" after they are used to wait for the trigger activation.  Triggers should be killed after they are no longer needed.  The best way, I think, is to kill them individually, but there will be cases where you want to kill all active triggers:

killTrigger triggerLabel         // kills the trigger named triggerLabel

killAllTriggers                       // kills all active triggers

When triggers are used in while/end loops, you can set them up so that you don't necessarily have to kill them individually.  The following is a snippet courtesy of ElderProphet for building a fig list showing how to reestablish triggers and make your code easier to read and more efficient:

# setup our label

 :buildFigArray

# setup our triggers to use to get the fighter list

setTextTrigger CorpFig :addCorpToList "Cor" 

setTextTrigger PersFig :addPersToList "Per" 

setTextTrigger ListDone :buildFigSec "Command" 

# setup pause to wait on the triggers

pause 

  

:addCorpToList                                                    // CorpFig trigger fired, others still active

getWord CURRENTLINE $figSector 1 

setVar $fighter[$figSector] $figSector 

add $corpCount 1 

setTextTrigger CorpFig :addToList "Cor"           // reestablish the CorpFig trigger 

pause                                                                    // pause

  

:addPersToList                                                     // PersFig trigger fired, others still active

getWord CURRENTLINE $figSector 1 

setVar $fighter[$figSector] $figSector 

add $persCount 1

setTextTrigger PersFig :addToList "Per"            // reestablish the PersFig trigger

pause  





     // pause

  

:buildFigSec                                                       // ListDone trigger fired, other still active

killTrigger CorpFig                                           // kill CorpFig trigger 

killTrigger PersFig                                            // kill PersFig trigger 

# tell us what we want to know

echo "*There are " $persCount " personal and " $corpCount " corporate." 

....

Aliens Causing Problems

Alien can cause a lot of script problems when they are hitting your grid.  Triggering off of a fighter hit on aliens wastes photons, ore, and moves if a turn game.  Not good for resource management and you aren't doing any good hitting aliens anyway with photons.

Sometimes though, you want to go after aliens so it is good to be able to tell an alien from a trader in that regard.  The easiest way I have found is with the use of CURRNTANSILINE.

  getWord CURRENTLINE $hit 5

  getWordPos CURRENTANSILINE $pos ": " & #27 & "[1;36m" & #27 & "[33m"           // alien color string

  getwordpos CURRENTANSILINE $pos2 ":" & #27 & "[1;36m"                                       // trader color string

  If ($pos > 0) or ($pos2 = 0)         // we want to make sure alien ANSI isn't there and trader ANSI is

       goto :resetScript

  end

Reading and Writing To A Text File

The .html text file does a good job of explaining how to read and write to a text file so I am going to use the coding example from it and break it down.

Read

# load bust list

# make sure that $BustFile contains a file of sector numbers to work with

if ($BustFile <> "")

  # see if the file name in $BustFile exists (from the TWX directory/folder)

  fileExists $exists $BustFile

  # We have checked the existance and stored it as True/False in $exists

  if ($exists)

    # setup a static array the size of SECTORS (stored in the TWX database)

    setArray $BustList SECTORS

    # init our line counter Variable 

    setVar $i 1

    # read the first line of the file

    read $BustFile $bust $i

    # while EOF (End Of File) does not = TRUE loop

    while ($bust <> EOF)

      # set our BustList to 1 so we know we have busted

      setVar $BustList[$bust] 1

      # increment our line counter

      add $i 1

      # read the new line at position $i

      read $BustFile $bust $i

    end       



// while loop – seeing if we have reached EOF

  end         



// making sure the file actually exists

end           



// making sure the file var contained text

Write

Writing to a file is very easy as TWX will either create a new file, or write to an existing one.  Sometimes you will want to delete a file prior to writing data to it (new fig list for example).

setVar $figList GAMENAME & ″.figs″          // setup a var containing the game name with .figs extension.

delete $figList                                                  // delete the old fighter list

setVar $i 1

while ($i <= SECTORS)

    write $figList   $deployedFig                     // write the sector number containing a fighter to the file

    add $i 1

end

Obviously the above is not all that would be involved in getting the fighter's sector numbers, but it shows how to setup a variable with a file name to write to, delete a file, and then write to the file.

Also note, that with the advent of setSectorParameter/getSectorParameter, the use of fighter lists is becoming a bit outdated.  The parameter set/get allows multiple scripts easy access to the data and allows your fighter list to be maintained in real time (unless a corpie is gridding).

